OPPOSITIONS OF MARS: NUMERICAL CALCULATION

Link to: physicspages home page.

To leave a comment or report an error, please use the auxiliary blog.

Reference: Carroll, Bradley W. & Ostlie, Dale A. (2007), *An Introduction to Modern Astrophysics*, 2nd Edition; Pearson Education - Chapter 2, Problem 2.18.

We can modify the Maple program that we used earlier for calculating orbits of planets to calculate the synodic period (time between successive oppositions). To do this, we need to calculate the orbits of two planets rather than just one, and we need to find the interval between the two times at which their angular coordinates are equal. To do this, we need to start out with both planets at $\theta_1 = \theta_2 = 0$ at $t = 0$ so they start out in opposition. If θ_1 is the angle of the inferior (closer to the Sun) planet, then this angle changes faster than θ_2, so we need to find the time t_{opp} when $\theta_1 = \theta_2 + 2\pi$.

There is one other complication however. In our original analysis of synodic periods, we assumed that the orbits are circular, so whenever one planet is in opposition to another, they are always the same distance apart. With elliptical orbits, that is no longer true. Oppositions can occur when the inferior planet is at aphelion and the superior planet is at perihelion (this gives the closest approach of the two planets) or vice versa (this gives the greatest opposition distance) or at some intermediate orientation. From the polar equation of an ellipse (with the principal focus at the origin):

$$r = \frac{a\left(1-e^2\right)}{1+e\cos\theta} \tag{1}$$

we see that perihelion occurs when $\theta = 0$ and aphelion when $\theta = \pi$. If we want the initial opposition to occur with the planets at some arbitrary positions in their orbits, we can add a phase to each angle, so that

$$r = \frac{a\left(1-e^2\right)}{1+e\cos\left(\theta+\theta_0\right)} \tag{2}$$

In the particular case of Earth and Mars, for example, we can start with Earth at aphelion ($\theta_{0E} = \pi$) and Mars at perihelion $\theta_{0M} = 0$. The next opposition will then occur when $\theta_E = \theta_M + 2\pi$.

A Maple program that calculates this for any two planets with arbitrary initial phases is

```
[code language="java"]
```

```
with(plots):
G := 0.6673e-10;
AU := 0.14959787066e12;
M__sun := 0.19891e31;
rad2deg := 180/Pi;
secsYear := 365.25*(3600*24);
orbit := proc (M__strsun, a__1AU, e__1, a__2AU, e__2, theta__01, theta__02, n)
local M__star, a__1, a__2, P__1, P__2, dt, t, theta__1, theta__2,
  LoM__1, LoM__2, r__1, r__2, i, dtheta__1, dtheta__2;
M__star := M__strsun*M__sun;
a__1 := a__1AU*AU;
a__2 := a__2AU*AU;
P__1 := sqrt(4*Pi^2*a__1^3/(G*M__star));
P__2 := sqrt(4*Pi^2*a__2^3/(G*M__star));
dt := 2*P__2/(n-1); t := Array(0 .. n, (i) -> i*dt);
r__1 := Array(0 .. n, datatype = float);
r__2 := Array(0 .. n, datatype = float);
theta__1 := Array(0 .. n, datatype = float);
theta__1[0] := theta__01;
theta__2 := Array(0 .. n, datatype = float);
theta__2[0] := theta__02;
LoM__1 := sqrt(G*M__star*a__1*(1-e__1^2));
LoM__2 := sqrt(G*M__star*a__2*(1-e__2^2));
for i from 0 to n-1 do
  r__1[i] := a__1*(1-e__1^2)/(1+e__1*cos(theta__1[i]));
  r__2[i] := a__2*(1-e__2^2)/(1+e__2*cos(theta__2[i]));
  dtheta__1 := LoM__1*dt/r__1[i]^2;
  dtheta__2 := LoM__2*dt/r__2[i]^2;
  theta__1[i+1] := evalf(theta__1[i]+dtheta__1);
  theta__2[i+1] := evalf(theta__2[i]+dtheta__2);
  if theta__2[i+1]+evalf(-theta__02+2*Pi) <= theta__1[i+1]-evalf(theta__01)
    and theta__1[i]-evalf(theta__01) < theta__2[i]+evalf(-theta__02+2*Pi) then
    print("Opposition at t = ", evalf(t[i]/secsYear));
    break
  end if
end do
end proc
[/code]
```

This code follows the same algorithm that we used earlier to calculate a single orbit, only this time we have two orbits to keep track of. The θ_1 and θ_2 arrays are initialized to θ_{01} and θ_{02} respectively and then incremented by $d\theta_{1,2}$ as usual. The **if** statement tests when the condition $\theta_E \geq \theta_M + 2\pi$ is first found and prints out the time (in years) when this occurs. Note that since the $\theta_{1,2}$ arrays start off with the initial phases, we need to subtract these phases in the **if** statement before doing the comparisons.

For Earth as planet 1 at aphelion ($\theta_1 = \pi$) and Mars as planet 2 at perihelion ($\theta_2 = 0$), we make the call

```
[code]
orbit(1, 1, 0.167e-1, 1.524, 0.934e-1, Pi, 0, 10000)
[/code]
```

which gives the result

$$t = 2.204 \text{ years} \tag{3}$$

If we put Earth at perihelion and Mars at aphelion, we make the call

```
[code]
orbit(1, 1, 0.167e-1, 1.524, 0.934e-1, 0, Pi, 10000)
[/code]
```

with the result

$$t = 2.091 \text{ years} \tag{4}$$

If we start Earth off at perihelion, in the slightly more than 2 complete orbits Earth makes before opposition occurs again, more of that time will be in the region of the orbit where Earth is closer to the Sun, where it moves faster, so we'd expect the time between oppositions to be less, as the calculations show.

For comparison, the result we got earlier assuming circular orbits gave $t = 2.111$ years, and the observed mean synodic period is 2.136 years. The two results we obtained from the numerical calculation should show the two extremes, so the answers appear to be about right.